
Executive Summary

IBNZAHR is considering six different operational setups for each of the three
reactors located at its main MBTE production facility. INT Consultants, Inc.
makes the following recommendation based on the criteria specified by IBN-
ZAHR: Reactor 1 should employ setup 1, Reactor 2 should employ setup 1, and
Reactor 3 should employ setup 6. This will result in an hourly production of 62
tons of MTBE.

Background

The Saudi European Petrochemical Company (IBNZAHR) contracted with INT
Consultants, Inc, to determine the optimum setup of the three reactors at the
MTBE production facility located in Al-Jubail. MTBE (methyl tert butyle
ether) is a popular additive in unleaded gasolines. MTBE offers an enviromen-
tally friendly alternative to the lead-based gasoline additives TEL (tetra ethyl
lead) and TML (tentra methyl lead). IBNZAHR currently produces 500,000
tons of MTBE per year. The production process is a hydrocarbon flow scheme
involving three reactors.

A mixture of butane and very small quantities of CT (carbon tetrachloride
and H2 (hydrogent) is fed into Reactor 1, known as the Isobutane reactor, which
converts the input mixture into isobutane. The isobutane is fed into Reactor
2, which is a dehydrogenation reactor that converts isobutane into isobutylene.
Only between 45 − 50% of the input isobutane gets converted into isobutylene
here, the remaining is recyled through this reactor again. The isobutylene is then
mixed with methanol and isobutane, the mixutre fed into Reactor 3, the MTBE
reactor, in which MTBE is produced by the methanol-isobutylene synthesis
process. Most of the input isobutane remains as it is at the end of processing in
Reactor 3, and it is recycled again. In each reactor, besides the main products,
some gas byproducts are also produced. These gase byproducts have potential
uses, but at the moment the plant does not have the necessary equipment to
process them, so they are being released into the atmosphere. These reactor
releases are currently air polutants.

Reactor 1 also produces liquid byproducts. They are also useful, but are not
being used currently. However, the liquid byproducts are disposed of safely and
hence do not contribute to pollution.

Each reactor can be operated at six different settings. The paramenters
which determine the reactor settings are the raw materials feed rate, the re-
actor pressure, and the reactor temperature. These parametere influence the
production rates, and the cost of operation, as shown in the tables in appendix
A. The cost of operation in each reactor includes the cost of any fresh raw ma-
terial inputs (i.e., butane, CT, and H2 in Reactor 1; methanol in Reactor 3.),
and energy and labor costs.

MTBE sells at the rate of 1332 SR/ton. And the plant has a limit of 0.665
tons/hr of gas byproduct releases from each reactor to limit air pollution.

Model Formulation

Let:
R The set of reactors
S The set of setups for each reactor
Oij Output of reactor i at setting j, for i ∈ R, j ∈ S
Iij Required materials input for reactor i at setting j, for i ∈ R, j ∈ S
Gij Gas by-product released from reactor i at setting j, for i ∈ R, j ∈ S
E Limit of gas byproduct released as pollution, per reactor
Cij Cost of setting reactor i at setting j, for i ∈ R, j ∈ S
P Selling price of MTBE per ton

xij

{
1, if reactor i operated at setting j

0, otherwise
, for i ∈ R, j ∈ S

The objective of the model is to maximize profit:

Maximize P ∗
∑
j∈S

O|R|,jx|R|,j −
∑
i∈R

∑
j∈S

Cijxij

A single setup must be selected for each reactor:∑
j∈S

xij = 1, for i ∈ R

The output of reactor i must match the input of reactor i + 1:∑
j∈S

Oi−1,jxi−1,j −
∑
j∈S

Ii,jxi,j ≥ 0, for i ∈ R

The selected set-up for each reactor must satisfy the required pollution limit:∑
j∈S

Gijxij ≤ E, for i ∈ R

The complete statement of the model is given as follows:
Maximize P ∗

∑
j∈S

O|R|,jx|R|,j −
∑
i∈R

∑
j∈S

Cijxij (maximize profit)

subject to
∑
j∈S

xij = 1, for i ∈ R (one setup per reactor)∑
j∈S

Oi−1,jxi−1,j −
∑
j∈S

Ii,jxi,j ≥ 0, for i ∈ R (reactor i produces enough for reactor (i + 1))∑
j∈S

Gijxij ≤ E, for i ∈ R (each reactor meets pollution standards)

xij ∈ {0, 1},∀i ∈ R, j ∈ S

Results and analysis

The model resulting from the provided information regarding the operating con-
ditions of each reactor is solved using the software AMPL (A Modeling Language
for Mathematical Programming), which is distributed by AMPL Optimization
LLC. The code used to generate the optimal solution is available in Appendix
B for review.

The solved model provides an optimal solution which calls for Reactor 1 to
employ setup 1, Reactor 2 to employ setup 1, and Reactor 3 to employ setup
6. Under this operational policy, IBNZAHR will produce MTBE at a rate of 62
tons per hour. If the plant is to be continually producing, this rate will result
in 1, 488 tons of MTBE daily. Gross profits are estimated to be $1, 982, 016.

However, even at the optimum production the daily net profit will be neg-
ative; it is estimated that IBNZAHR run a daily deficit of $1, 932. It is the
recommendation of INT Consultants, Inc. that IBNZAHR attempt to lower
operating costs, expand the MTBE market, or seek to utilize the gas and liquid
byproducts for profit.

Sensitivity analysis

A full sensitivity analysis was performed and the model was found to be stable.
The sensitivity data is available in Appendex C. The constraint relating to the
number of setups per reactor was found to be tight. That is, the optimum would
not be affected by weakening the integrality constraints. For this reason, INT
Consultants, Inc. recommends that once the optimum setup is performed, no
further changes be made in order to avoid incurring setup charges and a loss of
profit.

A - Reactor Operating Parameters

Reactor 1

Parameters Setup 1 Setup 2 Setup 3 Setup 4 Setup 5 Setup 6
Feed rates (tons/hr)
Butane 67 68 69 70 71 73
CT 0.009 0.0092 0.009 0.009 0.009 0.009
H2 0.065 0.067 0.069 0.070 0.075 0.066
Temp0C 164 165 166 168 170 169
Pressure (kg/cm2) 26 26 26 26 26 26
Isobutane output (tons/hr) 66.2 67.08 67.07 68.05 69.4 71.7
Byproduct gases (tons/hr) 0.4 0.42 0.43 0.45 0.6 0.5
Byproduct liquids (tons/hr) 0.4 0.5 0.6 0.7 1.0 0.8
Operating cost ( Sr/hr) 22,825 23149 23473 23797 24121 24769

*SR = Saudi Riyals

Reactor 2

Parameters Setup 1 Setup 2 Setup 3 Setup 4 Setup 5 Setup 6
Feed rates (tons/hr)
Isobutane 82.4 82.7 86.3 85.6 84.9 81.5
Temp0C 645 640 647 645 633 632
Pressure (kg/cm2) 30 30 30 30 30 30
Isobutane output (tons/hr) 42.34 40.23 42.75 40.55 39.62 33.89
Byproduct gases (tons/hr) 0.53 0.40 0.42 0.42 0.33 0.37
Remaining isobutane (tons/hr) 34.76 38.52 39.3 40.81 46.99 43.96
Operating cost ( Sr/hr) 33,747 33866 35284 35006 34725 33399

*Remaining isobutane is recycled through Reactor 2 again

Reactor 3

Parameters Setup 1 Setup 2 Setup 3 Setup 4 Setup 5 Setup 6
Feed rates (tons/hr)
Isobutylene 37 37.5 38 38.5 39 36
Methanol 21 21.2 21.5 22 23 24
Isobutane 34.8 38.5 39.3 40.8 48.6 44
Temp0C 65 66 67 68 70 71
Pressure (kg/cm2) 16 16 16 16 16 16
MTBE output (tons/hr) 57.8 58.4 59.1 60 61 62
Byproduct gases (tons/hr) 0.2 0.3 0.4 0.5 0.6 0.66
Byproduct liquids (tons/hr) 0.1 0.19 0.2 0.7 1.0 0.8
Remaining isobutane (tons/hr) 34.76 38.52 39.3 40.81 46.99 43.96
Operating cost ( Sr/hr) 25,813 26123 26461 26857 27397 26173

*Remaining isobutane is recycled again

B - AMPL Code

The AMPL code used to generate the optimal solution is available below, along
with the AMPL output.

AMPL Code
AMPL mod file:

param num_reactors = 3;

param num_setups = 6;

param Reactors {i in 1..num_reactors};

param Setups {i in 1..num_setups};

param Output {i in 1..num_reactors, j in 1..num_setups};

#materials outputed by reactor i with setup j, per hour

param Input {i in 1..num_reactors, j in 1..num_setups};

#Required materials input for reactor i with setup j, per hour

param Gas_Byproduct {i in 1..num_reactors, j in 1..num_setups};

#Gas byproducts produced by reactor i with setup j, per hour

param Enviromental_Restrictions = 0.665;

#Limit on the Gas byproducts released, per hour

param Cost {i in 1..num_reactors, j in 1..num_setups};

#cost to run reactor i with setup j, per hour

param Price = 1332;

#Selling price per ton

var x {i in 1..num_reactors, j in 1.. num_setups} binary;

#x[ij] = 1 if reactor i uses setup j and 0 otherwise

maximize Profits:

Price * sum{j in 1..num_setups} Output[num_reactors, j] * x[num_reactors, j] -

sum{i in 1..num_reactors, j in 1..num_setups} Cost[i,j]*x[i,j];

subject to One_Setup_Per_Reactor{i in 1..num_reactors}:

sum{j in 1..num_setups} x[i,j] = 1;

subject to Output_Matches_Input {i in 2..num_reactors}:

sum{j in 1..num_setups} Output[i - 1, j] * x[i-1, j] -

sum{j in 1..num_setups}Input[i,j]*x[i,j] >= 0;

subject to Pollution_Limits {i in 1..num_reactors}:

sum{j in 1..num_setups} Gas_Byproduct[i,j]*x[i,j] <= Enviromental_Restrictions;

5



AMPL dat file:

param Output: 1 2 3 4 5 6 :=

1 66.2 67.08 67.07 68.05 69.4 71.7

2 42.34 40.23 42.75 40.55 39.62 33.89

3 57.8 58.4 59.1 60 61 62;

param Input: 1 2 3 4 5 6 :=

1 67 68 69 70 71 73

2 47.64 44.18 47.0 44.79 37.9 37.6

3 37 37.5 38 38.5 39 36;

param Gas_Byproduct: 1 2 3 4 5 6 :=

1 0.4 0.42 0.43 0.45 0.6 0.5

2 0.53 0.40 0.42 0.42 0.33 0.37

3 0.2 0.3 0.4 0.5 0.6 0.66;

param Cost: 1 2 3 4 5 6 :=

1 22825 23149 23473 23797 24121 24769

2 33747 33866 35284 35006 34725 33399

3 25813 26123 26461 26857 27397 26173;

AMPL run file:

reset;

model mtbemod.mod;

data mtbedat.dat;

option solver gurobi;

solve;

display x;

display x >mtbe.sens;

display Profits >mtbe.sens;

display One_Setup_Per_Reactor.slack >mtbe.sens;

display Output_Matches_Input.slack >mtbe.sens;

display Pollution_Limits.slack >mtbe.sens;

AMPL Solution

ampl: include run.run;

Gurobi 5.5.0: optimal solution; objective -161

x :=

1 1 1

1 2 0

1 3 0

1 4 0

1 5 0

1 6 0

2 1 1

2 2 0

2 3 0

2 4 0

2 5 0

2 6 0

3 1 0

3 2 0

3 3 0

3 4 0

3 5 0

3 6 1

;

C - Sensitivity Analysis

The following information was obtained by performing a sensitivity analysis on
the model in AMPL.

Profits = -161

One_Setup_Per_Reactor.slack [*] :=

1 0

2 0

3 0

;

Output_Matches_Input.slack [*] :=

2 18.56

3 6.34

;

Pollution_Limits.slack [*] :=

1 0.265

2 0.135

3 0.005

;

7


	Executive Summary
	Background
	Model Formulation
	Results and analysis
	Sensitivity analysis

	A - Reactor Operating Parameters
	Reactor 1
	Reactor 2
	Reactor 3

	B - AMPL Code
	C - Sensitivity Analysis

